

Tutorial and examples for `tabout`

New users should jump ahead to the next section (Section 2) and then return to read Section 1. Existing users need only read the new information in Section 1.

1 Changes to `tabout` in Version 1.2.0

Two new options have been added to `-tabout-` in Version 1.2.0.¹ The `n` option allows you to specify the sample size (or population estimates) in a table; and the `nototals` option suppresses the repetition of the totals line in stacked tables and allows you to have just one totals line at the bottom of the table.

1.1 The `n` option

The `n` option allows you to include information on the sample size (or population estimates) in the body of the table. This can be useful when the table only contains percentages.

Here is an example of the `n` option in use:

```
tabout torque weight2 roomy foreign [iw=wt] using cars.txt, ///
      rep cells(row) format(%9.1f) n(col|d|d)
```

And here is the output produced:

¹The version numbers are not sequential, but move ahead in jumps. Do not be concerned if you do not have Version 1.1.9. This was a version for development purposes. The last publicly released version was Version 1.1.7.

Table 1: Various characteristics by origin

	<i>Manufacturing origin</i>			n
	Domestic origin	Foreign origin	Total	
<i>Category of gear ratio</i>				
Low	100.0	0.0	100.0	13
Middle	93.4	6.6	100.0	26
Upper	59.0	41.0	100.0	18
Highest	25.9	74.1	100.0	17
Total	71.9	28.1	100.0	74
<i>Weight category</i>				
Low	34.8	65.2	100.0	22
Medium	80.3	19.7	100.0	32
High	100.0	0.0	100.0	20
Total	71.9	28.1	100.0	74
<i>Roominess of the vehicle</i>				
Minimal headroom	91.6	8.4	100.0	4
Adequate headroom	63.6	36.4	100.0	55
Excellent headroom	100.0	0.0	100.0	15
Total	71.9	28.1	100.0	74

Source: Stata Dataset. Population: All cars.

Another variation on the n option is:

```

tabout torque weight2 roomy foreign [iw=wt] using cars.txt, ///
      rep cells(row) format(%9.1f) n(lab|(n = #)|d)

```

Which produces:

Table 2: Various characteristics by origin

	<i>Manufacturing origin</i>		Total
	Domestic origin	Foreign origin	
<i>Category of gear ratio</i>			
Low (n = 13)	100.0	0.0	100.0
Middle (n = 26)	93.4	6.6	100.0
Upper (n = 18)	59.0	41.0	100.0
Highest (n = 17)	25.9	74.1	100.0
Total (n = 74)	71.9	28.1	100.0
<i>Weight category</i>			
Low (n = 22)	34.8	65.2	100.0
Medium (n = 32)	80.3	19.7	100.0
High (n = 20)	100.0	0.0	100.0
Total (n = 74)	71.9	28.1	100.0
<i>Roominess of the vehicle</i>			
Minimal headroom (n = 4)	91.6	8.4	100.0
Adequate headroom (n = 55)	63.6	36.4	100.0
Excellent headroom (n = 15)	100.0	0.0	100.0
Total (n = 74)	71.9	28.1	100.0

Source: Stata Dataset. *Population:* All cars.

The syntax for the `n` option is: `n(pos label weight)`, with each sub-option separated by the pipe delimiter (or a delimiter of your own choice). All three sub-options must be specified. The first sub-option, `pos`, determines where the information on the `n` will appear in the table. There are two choice:

1. specifying `col` as the `pos` sub-option will make `-tabout-` append an extra column to your table and list the `n` counts in this column; or
2. specifying `lab` as the `pos` sub-option will make `-tabout-` include a label (such as `(n=125)`) following the value labels for the vertical variables

The form of words describing the `n` is specified in the `label` sub-option. Specifying `d` (for default) makes the column heading (if `col` has been chosen) show `"n"`. If you have specified `lab` in the `pos` sub-option, the default label is `"(n=#)"`. `-tabout-` replaces the `"#"` with the actual count in the final table. If you specify `lab` in the `pos` sub-option, you might notice that `-tabout-`'s screen display does not match the text file output. (Because the value labels are usually abbreviated for the screen display, there was no point in adding further words to the value labels. The extra column (when the `col` sub-option is chosen) does show in the screen display.)

If you do not want the default label, you can specify your own. Simply enter a phrase using the `"#"` symbol for where the actual count is to occur. For example, you might enter `"[count is #]"` or `"(sample size = #)"`.

The third sub-option specifies a weight to be used in calculating the sample size. The default (specified with `d`) is to use no weighting at all. This equates to a sample size count. If you wish to show population estimates, you might specify a weight variable, such as `wt` or `wght`. Note that `-tabout-` always uses Stata's `iweight` option for this weighting.

1.2 The nototals option

`nototals` suppresses the row totals which are repeated in the panels in stacked tables. In the case of row percentages, these totals are often the same and their repetition can be tedious and take up valuable page space. Issuing the `nototals` option removes all these row totals from the table.

Here is an example of the `nototals` option:

```
gen allcars=1
la var allcars "All cars"
```

```

la define allcars 1 "All cars"
la val allcars allcars

tabout torque weight2 roomy allcars foreign [iw=wt] using cars.txt, ///
      rep cells(row) format(%9.1f) nototals(allcars)

```

And here is the output produced:

Table 3: Various characteristics by origin

	<i>Manufacturing origin</i>		Total
	Domestic origin	Foreign origin	
<i>Category of gear ratio</i>			
Low	100.0	0.0	100.0
Middle	93.4	6.6	100.0
Upper	59.0	41.0	100.0
Highest	25.9	74.1	100.0
<i>Weight category</i>			
Low	34.8	65.2	100.0
Medium	80.3	19.7	100.0
High	100.0	0.0	100.0
<i>Roominess of the vehicle</i>			
Minimal headroom	91.6	8.4	100.0
Adequate headroom	63.6	36.4	100.0
Excellent headroom	100.0	0.0	100.0
<i>All cars</i>	71.9	28.1	100.0

Source: Stata Dataset. *Population:* All cars.

The `nototals` option must have a “total variable” inside its parenthesis. A “total variable” can be any variable you wish, even an “artificial” variable, as shown in the example above. It is called a “total variable” because it is used when you want to have a single total row, such as at the bottom of the stacked tables. To do this, add the “total variable” to your vertical variable list. Place it first in the list if you want it to display at the top of the table or place it last in the vertical list if you want it to display at the bottom. If you do not want any totals shown in the table, simply omit the “total variable” from your vertical variable list (but it must still appear inside the parenthesis for the `nototals` option).

The steps for implementing the `nototals` option are straightforward, if somewhat tedious. First create a “total variable”:

```
gen allcars = 1
```

and define some labels:

```
la var allcars ‘‘All cars’’ la def allcars 1 ‘‘All cars’’ la val allcars allcars
```

The variable label is what will be displayed, the value label is needed because `-tabout-` insists that all values are labelled.

Next include this variable in your vertical variable list as the last variable (if you want it at the bottom) or as the first variable (if you want it at the top). For example, to place it last:

```
tabout torque weight2 roomy allcars foreign
```

in a two-way table, and

```
tabout torque weight2 roomy allcars, oneway
```

in a one-way (frequency) table.

For housekeeping reasons, `-tabout-` makes use of @@ symbols on the line where the total variable will be shown. These do not appear in the final table. If for some reason, you have used two @@ symbols in any of your labels, you may have problems. Simply change your own @@ symbols and everything will be fine.

2 Examples of tabout tables

2.1 What tabout does

`-tabout-` is a *Stata* ado program for producing publication quality tables. The output produced is either L^AT_EX (or TeX), HTML or tab-delimited. `-tabout-` can produce:

- frequency counts, percentages and cumulative percentages (similar to `-tabulate-`);
- counts, row percentages or column percentages in cross-tabulations (also similar to `-tabulate-`);
- summary statistics in either cross-tabulations or in one-way tables (similar to `-table-`, but with additional statistics such as percentile ratios, for example, the 90th to the 10th);

In addition `-tabout-` can ‘stack’ tables into a larger table, as a series of panels, one below the other.

2.2 Where can it be obtained?

`-tabout-` is available from the SSC archives. From inside *Stata* just type: `ssc install tabout`. The current version is 1.2.0. Note that the syntax is different in a number of areas to the syntax used in version 1.0.0.

There are two main files: `tabout.ado` and `tabout.hlp`. In addition, there are some support files:

- `example_auto.do` which was used to generate some of the tables in this document.
- `htmltables.html` and `taboutexamples.pdf` (this document), both available from: <http://www.acirrt.com/watson/tabout>

2.3 Using *Stata*'s auto dataset

The tables shown below all make use of the standard *Stata* dataset, `auto.dta`. To make the data more suitable for cross-tabulations and to better illustrate the feature of `-tabout-`, the following modifications are carried out first:

```
sysuse auto, clear
recode weight (min/2499=1 "Low") ///
              (2500/3499=2 "Medium") ///
              (3500/max=3 "High"), ///
              gen(weight2)
la var weight2 "Weight category"
recode gear_ratio (min/2.49=1 "Low") ///
                (2.5/2.99=2 "Middle") ///
                (3/3.49=3 "Upper") ///
                (3.5/max=4 "Highest"), gen(torque)

la var torque "Category of gear ratio"
la def foreign 0 "Domestic origin" 1 "Foreign origin", modify
la val foreign foreign
la var foreign "Manufacturing origin"
recode headroom (min/1.9=1 "Minimal headroom") ///
                (2/3.9=2 "Adequate headroom") ///
                (4/max=3 "Excellent headroom"), gen(roomy)
la var roomy "Roominess of the vehicle"
gen wt=uniform()*1000
la var weight "Weight"
la var mpg "Mileage"
la var length "Length"
la var displacement "Capacity"
```

2.4 Frequency tables

`-tabout-` can produce simple frequency tables identical to those in `-tabulate-`:

A table like the one which follows can be produced with as little code as:

```
tabout torque [iw=wt] using cars.txt, ///
replace cells(fcount fper fcum) format(%9.0fc %9.1f %9.2f)
```

Note the `cells` option specifies the contents of the cells in the table, and the `format` option provides different formats for each column.

Table 4: Gear ratio, simple frequencies

Category of gear ratio	Number	Per cent	Cum Per cent
Low	6,564	18.6	18.58
Middle	12,284	34.8	53.36
Upper	9,341	26.4	79.81
Highest	7,133	20.2	100.00
Total	35,322	100.0	.

Source: Stata Dataset. *Population:* All cars.

Most of the other enhancements to the table simply involve a few more additions. The core of the syntax is covered in this example.

If you want to stack tables, just add more variable names. Note, however, the addition of the `oneway` option if you want them all to be frequency tables. If you neglect this option, `-tabout-` assumes the last variable in your list is a ‘horizontal’ variable, and it produces cross-tabulations.

Note the difference:

```
tabout torque weight2 roomy foreign [iw=wt] using cars.txt, ///
      rep cells(fcount fper fcum) format(%9.0fc %9.1f %9.2f) oneway
```

produces:

Table 5: Various characteristics, simple frequencies

	Number	Per cent	Cum Per cent
Category of gear ratio			
Low	6,564	18.6	18.58
Middle	12,284	34.8	53.36
Upper	9,341	26.4	79.81
Highest	7,133	20.2	100.00
Total	35,322	100.0	.
Weight category			
Low	10,368	29.4	29.35
Medium	16,105	45.6	74.95
High	8,849	25.1	100.00
Total	35,322	100.0	.
Roominess of the vehicle			
Minimal headroom	2,676	7.6	7.58
Adequate headroom	26,670	75.5	83.08
Excellent headroom	5,975	16.9	100.00
Total	35,322	100.0	.
Manufacturing origin			
Domestic origin	25,394	71.9	71.89
Foreign origin	9,928	28.1	100.00
Total	35,322	100.0	.

Source: Stata Dataset. Population: All cars.

Whereas:

```
tabout torque weight2 roomy foreign [iw=wt] using cars.txt, ///  
  rep cells(fcount fper fcum) format(%9.0fc %9.1f %9.2f)
```

produces an unintended:

Table 6: Various characteristics, simple frequencies

	Manufacturing origin		Total
	Domestic origin	Foreign origin	
Category of gear ratio			
Low	6,564	0	6,564
Middle	11,475	809	12,284
Upper	5,510	3,832	9,341
Highest	1,845	5,287	7,133
Total	25,394	9,928	35,322
Weight category			
Low	3,613	6,755	10,368
Medium	12,932	3,173	16,105
High	8,849	0	8,849
Total	25,394	9,928	35,322
Roominess of the vehicle			
Minimal headroom	2,450	226	2,676
Adequate headroom	16,968	9,703	26,670
Excellent headroom	5,975	0	5,975
Total	25,394	9,928	35,322

Source: Stata Dataset. Population: All cars.

2.5 Cross-tabulations

A table like this:

Table 7: Gear ratio by origin

Category of gear ratio	Manufacturing origin		
	Domestic origin	Foreign origin	All cars
Low	100.0	0.0	100.0
Middle	93.4	6.6	100.0
Upper	59.0	41.0	100.0
Highest	25.9	74.1	100.0
All categories	71.9	28.1	100.0

Source: Stata Dataset. *Population:* All cars.

is produced with this syntax:

```
tabout torque foreign [iw=wt] using cars.txt, ///
    rep c(row) f(%9.1f) total(All categories|All cars)
```

Note the use of the `total` option. This allow you to customise how totals are shown in your tables. The first term in the option is for the ‘vertical’ variable(s), and the second term for the ‘horizontal’ variable. The terms may have spaces in them, but note the use of the pipe delimiter. If you wish to use another symbol, you can specify this in the `delim()` option. For example, `delim(*)` would then mean you enter `total(All categories*All cars)`.

You can also present chi-square statistics beneath the table, provided it is a simple two-way cross tabulation (that is, not stacked). If users find this limitation too restrictive, future versions of `-tabout-` could allow for the chi-square option within stacked tables. I would appreciate hearing from users if this is desirable, and the best manner of presentation.

Note also that `iweights` and `aweight`s cannot be used. This is a *Stata* restriction (see `help tabulate` for more information).

Table 8: Gear ratio by origin, with chi-square results

Category of gear ratio	Manufacturing origin		Total
	Domestic origin	Foreign origin	
Low	13	0	13
Middle	25	1	26
Upper	11	7	18
Highest	3	14	17
Total	52	22	74

Pearson chi-square (3 df) = 37.09, probability = 0.000

Source: Stata Dataset. *Population:* All cars.

The syntax for this is:

```
tabout torque foreign using cars.txt, ///
  rep c(count) f(%9.0f) ///
  stats(chi2 2 3)
```

You must type exactly the term `chi2`, and not use abbreviations (as you can in `-tabulate-`). You need not type in anything else, and `-tabout-` will use 3 decimal points as its default. Otherwise, as shown here, you can type in the number of decimal points for the chi-square value and the p value. Note that the terms `c(count)` and `f(%9.0f)` are redundant here, since these are the defaults for cross-tabulations.

With cross-tabulations, you can stack the tables like this:

Table 9: Various characteristics by origin

	<i>Manufacturing origin</i>		Total
	Domestic origin	Foreign origin	
<i>Category of gear ratio</i>			
Low	100.0	0.0	100.0
Middle	93.4	6.6	100.0
Upper	59.0	41.0	100.0
Highest	25.9	74.1	100.0
Total	71.9	28.1	100.0
<i>Weight category</i>			
Low	34.8	65.2	100.0
Medium	80.3	19.7	100.0
High	100.0	0.0	100.0
Total	71.9	28.1	100.0
<i>Roominess of the vehicle</i>			
Minimal headroom	91.6	8.4	100.0
Adequate headroom	63.6	36.4	100.0
Excellent headroom	100.0	0.0	100.0
Total	71.9	28.1	100.0

Source: Stata Dataset. Population: All cars.

The main change in the syntax to achieve this kind of stacking involves adding more variables to the varlist. Note that the ‘horizontal’ variable is always the last one in the list. As mentioned earlier, the `oneway` option negates this last ‘horizontal’ variable, and treats all of the variables as ‘vertical’.

```
tabout torque weight2 roomy foreign [iw=wt] using cars.txt, ///
    rep c(row) f(%9.1f)
```

With cross-tabulations like these, you only specify one `cells` option: either `count`, `row` or `col`, the latter two being for percentages. Unlike *Stata*’s `-tabulate-` command, `-tabout-` does not allow for multiple contents in cells. There is one exception to this. You can specify the `double` option in cells and produce a table like this:

Table 10: Various characteristics by origin

	Manufacturing origin					
	Domestic origin		Foreign origin		Total	
	Num	%	Num	%	Num	%
Category of gear ratio						
Low	6,564	25.8	0	0.0	6,564	18.6
Middle	11,475	45.2	809	8.1	12,284	34.8
Upper	5,510	21.7	3,832	38.6	9,341	26.4
Highest	1,845	7.3	5,287	53.3	7,133	20.2
Total	25,394	100.0	9,928	100.0	35,322	100.0
Weight category						
Low	3,613	14.2	6,755	68.0	10,368	29.4
Medium	12,932	50.9	3,173	32.0	16,105	45.6
High	8,849	34.8	0	0.0	8,849	25.1
Total	25,394	100.0	9,928	100.0	35,322	100.0
Roominess of the vehicle						
Minimal headroom	2,450	9.6	226	2.3	2,676	7.6
Adequate headroom	16,968	66.8	9,703	97.7	26,670	75.5
Excellent headroom	5,975	23.5	0	0.0	5,975	16.9
Total	25,394	100.0	9,928	100.0	35,322	100.0

Source: Stata Dataset. Population: All cars.

2.6 Summary statistics

`-tabout-` can produce tables of summary statistics, such as means, medians, standard deviations, interquartile ranges, various percentiles, and so forth, and place these in cross-tabulations. In this way, it is like *Stata's* `-table-` command. `-tabout-` can also produce other statistics such as percentile ratios: 90th to 10th; 90th to 50th; 75th to 25th; and 10th to 50th.

It is important to note that `-tabout-` use *Stata's* `-summarize-` command and requires that you use `aweight`s, and not `iweight`s, if you require weighted results. This makes no difference to your results, only to the syntax you type.

Here is an example of a cross-tabulation in which the contents of the cells are summary statistics (the mean of mpg):

Table 11: Average fuel consumption by origin

	Manufacturing origin		Total (mean mpg)
	Domestic origin (mean mpg)	Foreign origin (mean mpg)	
Category of gear ratio			
Low	16.0	.	16.0
Middle	20.6	17.0	20.4
Upper	21.1	23.5	22.1
Highest	27.8	24.7	25.5
Total	20.0	23.6	21.0
Weight category			
Low	28.7	26.7	27.4
Medium	20.5	17.0	19.8
High	15.8	.	15.8
Total	20.0	23.6	21.0
Roominess of the vehicle			
Minimal headroom	23.4	23.0	23.4
Adequate headroom	20.2	23.6	21.4
Excellent headroom	18.1	.	18.1
Total	20.0	23.6	21.0

Source: Stata Dataset. Population: All cars.

It was produced with the following syntax:

```
tabout torque weight2 roomy foreign [aw=wt] using cars.txt, ///
  rep c(mean mpg) f(%9.1f) ///
  header3(_|(mean mpg)|(mean mpg)|(mean mpg))
```

Note the change from `[iw=wt]` to `[aw=wt]` and the specification of `mean mpg`. If you wanted the median, you could specify either `p50 mpg` or `median mpg`, `-tabout-` understands both. Whereas `-tabout-` automatically enters variable labels and value labels into headers, it does not attempt to do this with the summary statistics, since the required wording is difficult to anticipate. However, by using the `header3` option (as in this example), you can provide the additional information in the header of the table. I'll discuss headers at length below.

Summary statistics can also be used in `oneway` tables, and in this way different statistics can be nominated. You do not need to specify `oneway` because `-tabout-` recognises that a oneway table is required when it finds multiple statistics nominated in the `cells` option. For example:

Table 12: Summary statistics according to gear ratio

	Various characteristics of interest			
	Mileage (mn mpg)	Weight (med lbs)	Length (sd in)	Capacity (max cu in)
Category of gear ratio				
Low	16.0	4030	11.43	425
Middle	20.4	3210	17.77	304
Upper	22.1	2650	18.22	231
Highest	25.5	2120	15.51	163
Total	21.0	3200	21.97	425
Weight category				
Low	27.4	2110	9.85	304
Medium	19.8	3210	10.45	258
High	15.8	4030	9.27	425
Total	21.0	3200	21.97	425
Roominess of the vehicle				
Minimal headroom	23.4	2520	23.48	231
Adequate headroom	21.4	2750	21.09	400
Excellent headroom	18.1	3670	12.66	425
Total	21.0	3200	21.97	425
Manufacturing origin				
Domestic origin	20.0	3350	20.23	425
Foreign origin	23.6	2200	15.45	163
Total	21.0	3200	21.97	425

Source: Stata Dataset. Population: All cars.

is produced with:

```
tabout torque weight2 roomy foreign [aw=wt] using cars.txt, rep ///
  c(mean mpg p50 weight sd length max disp) ///
  f(%9.1f %9.0f %9.2f %9.0f) ///
  header1(_|Various characteristics of interest) ///
  header3(_|(mn mpg)|(med lbs)|(sd in)|(max cu in)) ///
```

Note the addition of `header1`. As will be explained shortly, you can mix your own headers with the pre-existing (automatically generated) headers. Note also that you can specify as many formats as there are columns. If you specify more formats than columns, the extra ones are ignored. If you specify fewer formats than columns, then you may find some of your columns are displayed to 6 decimal points.

As mentioned earlier, `-tabout-` also produces tables of percentile ratios, such as the 90th to the 10th. These are specified with an `r` in front of them, such as: `r9010`. For example:

Table 13: Summary statistics according to gear ratio

	Distribution aspects of car prices			
	iqr	90/10 ratio	90/50 ratio	10/50 ratio
Category of gear ratio				
Low	7301	3.0	1.3	0.4
Middle	926	2.8	2.2	0.8
Upper	852	1.5	1.3	0.9
Highest	3156	3.4	2.6	0.7
Total	2305	3.0	2.4	0.8
Weight category				
Low	1184	1.7	1.4	0.9
Medium	1041	3.0	2.5	0.9
High	5332	2.5	1.5	0.6
Total	2305	3.0	2.4	0.8
Roominess of the vehicle				
Minimal headroom	2097	1.5	1.3	0.9
Adequate headroom	3058	3.4	2.8	0.8
Excellent headroom	3094	2.2	1.5	0.7
Total	2305	3.0	2.4	0.8
Manufacturing origin				
Domestic origin	2161	2.6	2.2	0.8
Foreign origin	3145	3.2	2.1	0.7
Total	2305	3.0	2.4	0.8

Source: Stata Dataset. Population: All cars.

This was produced with this syntax:

```
tabout torque weight2 roomy foreign [aw=wt] using cars.txt, rep ///
  c(iqr price r9010 price r9050 price r1050 price) ///
  f(%9.0f %9.1f %9.1f %9.1f) ///
  header1(_|Distribution aspects of car prices) ///
  header2(_|iqr|90/10 ratio|90/50 ratio|10/50 ratio) ///
```

Compare these last two tables and note the use of `header2` in this last example to replace the pre-existing header (which would have printed the variable label, “price”).

3 The output produced by `tabout`

3.1 Users of Word Processors and Spreadsheets

The default output from `-tabout-` is tab delimited text. You can also produce \LaTeX (or \TeX) output with the `style(tex)` option, and html output with the `style(html)` option. The output is sent to the file specified in `using` and can be further processed according to your needs. If you wish to view the output code on the screen, then add `show(code)` or `show(all)` to your syntax for `-tabout-`.

By default the table is displayed on the screen but you can switch this off with `show(none)`, in which case you will only see a “file written” message. Sometimes, the display of the headers on the screen may not look correct, and will often appear abbreviated. Do not worry, the file output will always be correct, since the labels and various names are never abbreviated in the file output. The screen display is only intended to give you a general idea of what the table will look like.

There appears to be a quirk in *Stata* with copying and pasting from tab delimited text from the screen into spreadsheets and word processors. Sometimes it seems to work properly, sometimes it doesn't. Copying and pasting from the text file created by `-tabout-` works fine, so if you have problems copying from the screen, open your text editor and copy from there.

There are several way to get the tab delimited output into word processors and spreadsheets like *Word* and *Excel*:

- For *Excel*:
 1. you can copy and paste the delimited text straight into the cells.
 2. alternatively, you can use *Excel*'s **File Open** command and import your text file. The text file that you open is the filename specified in the **using** part of the `-tabout-` syntax. Note that if you use a `.txt` extension to your file (eg. `cars.txt`), then *Excel* automatically recognises that this is a text file, and that it is tab delimited. The steps you use for this procedure are quite simple:
 - **File Open**, then type in `cars.txt` under **File name**, (or you can select **All Files** or **Text Files** in the **Files of Type** list, and then click on `cars.txt` when it appears).
 - *Excel*'s **File Import Wizard** will start automatically. Just answer **Next Next Finish** and the output from `-tabout-` will end up in the correct columns without you needing to provide any information.
- For *Word*:
 1. you can copy and paste the delimited text straight into an open *Word* document, select the text and use **Table / Convert / Text to Table** to place it into a table;
 2. you can open your text file, for example `cars.txt`, within *Word* using **File Open**. You can either type in `cars.txt`, or you can select **All Files** or **Text Files** in the **Files of Type** list, and then click on `cars.txt` when it appears. If you get any queries about the type of file, just click on **OK**. You should then see the text displayed with tabs. If you cannot see the tabs, you may need to go **Tools / Options / View** and tick the **Tab characters** box. At this point, you can just select the text and use **Table / Convert / Text to Table** to place the text into a table;
 3. you can also output the tables from `-tabout-` as html, open them in your browser, and copy and paste them into *Word*, using the **Paste Special** command, and selecting **Formatted Text (rtf)**.


```

\footnotesize
\begin{spacing}{0.9}
\parbox{\mywide}{\emph{Source:} #.
                \emph{Population:} #}
\end{spacing}
\end{table}

```

Comparing the code, you can see how `-tabout-` replaced the `#` symbol with the arguments passed by your syntax. If you want to use a different symbol, indicate this with `psymb`, and if you don't want to pass any arguments to these `.tex` files, just leave out the `psymb` option. If you want to use a different delimiter within the strings, use the `delim` option to specify this, for example, `delim(~)`. As mentioned earlier, this also applies to other parts of `-tabout-` which make use of delimiters, such as the headers and the total labels.

Words of warning when passing arguments: the length of your strings are limited by *Stata's* limits: 80 characters for *Intercooled Stata* and 244 characters for *Stata SE*. These limits apply not just to the argument, but to the total string. In the last example it would include `\emph{Population:}` as well as the argument. Also note that each placemaker symbol `#` must be on a new line in the `topf()` and `botf()` files. This is essential.

3.3 Using `tabout` with HTML

If you want to produce html tables, you could consider using the `append` option and gathering all of the tables into one html file, which could then be displayed in your browser in its entirety. You could specify a `topf()` at the beginning of your syntax, and a `botf()` at the end, in this way sandwiching all of the tables between the appropriate html top and bottom code. The example file, `htmltables.html` was produced in this way.

If you specify the `body` option, `-tabout-` automatically inserts either some basic html (or basic \LaTeX code) above and below your table. This allows you to view the appearance of the tables straight away, without further coding required. Of course, you are more likely to get the kind of results you prefer by creating your own `topf()` and `botf()` files and making use of these.

The HTML code exported by `-tabout-` is fairly basic, and could probably be improved inside `-tabout-` if necessary. Hopefully, the main improvements could come from appropriate HTML code inside the `topf()` and `botf()` files. The HTML examples in `htmltables.html` make use of two `topf()`

and `botf()` files. The first one (`main_top.htm`) is as follows:

```
<HTML>
<BODY>
<H1>#</H1>
<P>
</P>
<H2>#</H2>
<TABLE>
```

This is attached to the first table in the series. All the subsequent tables are appended to this file, and they are interspersed with the other `topf()` file (`example_top.htm`), as follows:

```
<P>
</P>
<H2>#</H2>
<TABLE>
```

Each of these ‘middle’ tables has the following very simple `botf()` file (`example_bot.htm`) attached:

```
</TABLE>
```

while the final table has the following file (`main_bot.htm`):

```
</TABLE>
</BODY>
</HTML>
```

The following snippet of `-tabout-` code shows how the system works:

```
tabout torque [iw=wt] using cars.htm, rep style(html) ///
  c(fcount fper fcum) f(%9.0fc %9.1f %9.2f) font(bold) ///
  topstr(Examples of HTML tables using tabout|Gear ratio, ///
  simple frequencies) ///
  psymb(#) topf(main_top.htm) ///
  botf(example_bot.htm)
```

4 Headers

Headers are the most complicated part of `-tabout-` (and the most complex part of the programming). The default headers work reasonably well and should suffice for most occasions. The principles are as follows:

- there are at most three headers, though only two are used in the default settings (except for the `double` type of table, which makes use of all three).
- the first line of the headers begins above the numerical cells of the table. This is usually the ‘horizontal’ variables label. It is usually fonted (bold or italic) in \LaTeX or HTML.
- the second line of the headers usually contains the values of the ‘horizontal’ variable. These also begin above the numerical cells.
- as can be seen above, the variable labels for the ‘vertical’ variables are usually below the headers. An exception to this is simple tables (that is, those that not stacked), which place the single ‘vertical’ variable’s label in the first column of the bottom row of the headers (similar to what `-tabulate-` does). These types of tables also place a rule above the Total line in the table, in the style of the `-tabulate-` command. In \LaTeX and HTML these labels are also usually fonted.

When it comes time to add your own headers, you can simply overwrite the line you wish to change, and the other lines will remain unaffected. If you wish your header to start in the second column (that is, above the numerical cells), enter an underscore, followed by the pipe (or other) delimiter into your string, for example:

```
header1(_|This is my main heading)
```

If you want to skip any other columns, also use underscores in this way, for example:

```
header3(_|First numerical column|_|Now third one|_|Now fifth one)
```

The user headers work well using the `tex` style, though you need to enter your own code for `\multicolumn` spans, and for fonts. The user headers do not work as well with HTML because you often need to control the following line, that is, `header1` content can be easily changed, but `header2` may also have to be changed. Possibly the easiest option here is to specify the `noheader` option, which suppresses all automatic header output, and then create your own custom commands in the `topf()` file for the table.

Sometimes, the only change needed to the automatic headers is a desire to change the location of the ‘vertical’ variable’s label (which appears in the first column, usually on the bottom row of the headers). Rather than having to create a user header (which has implications for the whole line, not just the first column), you have another strategy: the `lefthead` option takes the values 1, 2 or 3, and these determine on which of the three header lines this label appears. Re-positioning the label in this way has no effect on the rest of the header, that is, the labels which are above the numerical cells.

5 Final words

`-tabout-` works well in \LaTeX , under which it was developed. Nevertheless, it can no doubt be improved, and I would welcome suggestions. `-tabout-` can certainly be improved for HTML, and I would also welcome advice on this. As mentioned earlier, copying the tab delimited output from the screen in *Stata* into spreadsheets and word processors is uneven in its results. Inspection of the tab delimited files in a text editor shows all of the tabs in the correct places. I would appreciate feedback from users for any improvements in this area.

If you have problems with `-tabout-` the first thing to check is that all your variables and values are labeled. This is essential for `-tabout-` to work, and you should get a warning message if you try to use string variables or unlabeled numeric variables. You may also have problems if any of your variables use a value of `-9`, or if any of your labels use underscores (`_`). `-tabout-` uses both `-9` and `_` for its own housekeeping. The simplest strategy is for you to remove the underscores from your labels (note, that I am not talking about variable names, only labels, as well as value labels), and for you to avoid `-9` in your values. If the latter is too onerous, you can easily modify the code for `-tabout-` and replace `-9` with another obscure number. If you are using *Intercooled Stata*, you might also have problems if you are constructing large tables and your value labels are very long. This may happen because

`-tabout-` processes tables using strings, and *Intercooled Stata* is restricted to strings of length 80 characters. Please contact me if you do run into problems with this limit.

Finally, as mentioned before, do not worry too much about the visual display of the table: the text files will be correct even if the columns on the screen look odd. You can improve this display with the `tabwidth`, `colwidth` and `labwidth` options. These have no effect on the output sent to the file, but they can make the screen display more pleasing.

Ian Watson
acirrt
University of Sydney
Australia
i.watson@econ.usyd.edu.au

13 March 2005
Current version of `-tabout-`: 1.2.0